
Introduction to VBScript by Guy Thomas 

The scripts in this collection of ten Ezines are designed to give you fun while 
building VBScripts.  As a bonus, one day you may actually find a practical 
application for one of my 'missions'.  Meanwhile, collecting the VBScript 
methods and commands will help you understand and troubleshoot your own 
scripts.  These 10 Ezines were originally numbered 97 - 107 and were first 
published in the spring of 2006. 

My style of teaching is to learn by doing.  Occasionally theory rears its head 
in my Ezines, but only to explain where we need to see patterns in the code.  
If I was to analyze my niche it would be to bridge that gap between the 
complete philistine who just copies and pastes scripts without understanding, 
and those who want structured theoretical teaching. 

The rhythm of the Ezines is as follows: a clear mission with a short 
explanation of how to achieve our goals.  In the middle is an example script 
containing the code itself, followed by learning points for you to understand 
the purpose of the commands.  These particular Ezines also have challenges 
whereby you can learn through altering my code to suit your environment. 

Script Advice and Help 

I run a support service where I charge $25 to help people with their scripts.  
Here are three free pieces of advice which underpin 80% of all VBScript 
problems. 

1) Build up your script in manageable stages.  Or if you are troubleshooting, 
break the script down into workable modules.  Get each part working then 
bolt all the sections together. 

2) Remember that scripts mimic what you do manually, so have a 
walkthrough using the interfaces that your script is trying to recreate. 

3) If you do get an error play close attention to the line number, this 
reference is your biggest clue in solving the problem.  If you still cannot 
figure out the error, look up the 800x0000 error message in Google. 

I will let you into a secret. 

I would prefer not to run my support service.  What I would really like is for 
people to solve their own problems, that is why I tell you all I know in the 
ezine or on the web site.The main reason I offer this service is because I 
owe it to my readers to assist them, especially if it's an easy problem.  A 
second reason I run the service is because I like helping people.   

Now the reason that I don't want to operate the service is that it's time 
consuming, some thorny problems take days to solve and for $25 that's not 
cost effective.  While the problems don't take up every hour of the day, I still 
cannot rest until I solve the problem and this prevents me concentrating on 
other jobs. 

VBScript Topics 

Here are the two parallel strands:  

 Tasks and Missions 
for VBScript 

1. Time setsntp

 

2. W32tm

 

3. Restart Services

 

4. If... then End If 
Statement

 

5. Binding to Active 
Directory

 

6. Methods

 

7. Create Users

 

8. File System Object 
(FSO)

 

9. Arguments

 

10. Error Correcting 
Code

   
VBScript Methods, Commands and Syntax 

 
' Comments (1) 

 

DIM Statements (2) 

 

error-correcting Code (10) 

 

If Then.. Else... End If  (4) 

 

Loop For...Next  (6) 

 

On Error Resume Next (10) 

 

Option Explicit (1, 2) 

 

Select Case (10) 

 

Sub Routine (2) 

 

WScript.Echo (1-10) 

 

Underscore (_) (9) 

 

Variables - Hungarian Convention (2)   

 

Bind to Active Directory 
GetObject("LDAP://RootDSE") (5) 

 

.Create("Organizational Unit")  .SetInfo (7) 

 

CreateObject("WScript.Network") (9) 

 

CreateObject("WScript.Shell") (7,10) 

 

CreateObject("Scripting.FileSystemObject") 
(8) 

 

MapNetworkDrive  (9)    

Who is this Introduction to VBScript ebook aimed at? 

 

Administrators who need to write simple logon scripts. 

 

Techies wishing to automate repetitive tasks. 

 

Script writers who are curious to know the difference between VBScript 
and JScript. 

 

Network Mangers who to more control over their servers.  

Contents for Ezine 1 - Net Time /setsntp 

Introduction to 2006 Ezines 

The New Year is a classic time for fresh initiatives, coupled with reflections 
on the previous 12 months.  As far as this ezine is concerned, the new 
initiative is a 10 part series specializing in the basis and basics of VBScript.  
As far as looking to the past, I just want to say that this ezine will continue 
with the twin aims of providing working script examples, while giving general 
learning points to help you adapt my scripts to your environment.   

 

This Week's Secret

  

This Week's Mission

  

Pure Scripting Learning Points

  

Example: VBScript to set the Time Server

  

Summary

 

This Week's Secret 

About five years ago, when this site was just a few pages, I noticed that the 
Logon Script page had more hits than all the rest of the pages added 
together.  Realising the interest in VBScript for Windows, I developed a 
whole scripting section.  Two years ago, I started this ezine primarily as a 
way of giving those who bought my ebooks an after sales service. 

From the first ezine, my specialty has been getting people started.  So for 
the next ten editions I am going to concentrate on the basics of VBScript.  I 
will be choosing real world examples for you to copy and paste, but my real 
objective is to explain how scripts are put together.  In particular, I want to 
explain which elements of the script you can and should change, and which 
elements are fixed terms have fixed names that you should learn, apply, but 
never change. 

This Week's Mission 

This week's mission is to configure your machine so that it synchronizes with 
an internet time server.  Trust me, accurate time is important on a server.  
Moreover, no-one can explain, to my satisfaction, why left to its own 
devices, my $1,000 computer regularly loses nearly a minute every week.  
Especially when my 10 year old, $5, 'Mickey Mouse' watch keeps perfect 
time. 

Theory that it's helpful to know. 

1. Windows operating systems have a built-in 'net time' command.   
2. SNTP is the simple network time protocol, not to be confused with 

SMTP.   
3. There are a number of publicly available time servers on the internet 

that are kept accurate to the nearest nanosecond, for example, at NIST 
(National Institute of Standards and Technology). 

Contents for Ezine 3 - Restart Services 

 
This Week's Secret

  
This Week's Operating System Mission -  Restart Services

  
This Week's Scripting Mission -  Variables and Sub Routines

  
Example: To Restart the Windows Time Service

  
Summary -  Restarting the Time Service

 

This Week's Secret 

It's an open secret that scripters need to be multi-talented. Not only do we 
need knowledge of VBScript, but also understanding of the operating 
system's mechanics.  A generous dose of logic is also useful for greasing the 
scripting wheels. 

As ever, I want to combine a script that is useful in its own right, with code 
that can be adapted to other scenarios.  This week's script, which restarts 
Windows services fits the bill perfectly. 

As you may know, I use the OnScript editor for my own scripts.

 

 In 
February 2006 OnScript introduced a new affiliate program which gives me a 
% of the sale rather than a fixed price for advertising in this ezine.  To get 
an evaluation copy use any of my links, then press the 'Back' button once 
you reach OnScript's site. 

This Week's Operating System Mission - Restart Services 

As a matter of tactics, it's much more satisfying to cure a problem by 
restarting a service than by rebooting the machine.  Not only is restarting 
the service faster and less disruptive for users, but also it teaches you to 
pinpoint the problem. 

This Week's Scripting Mission - Variables 

I want to introduce you to the benefits of variables in scripts.  I also want to 
explain how I employ an adaptation of the Hungarian Variable system in my 
scripts.  Most people find a good script and then use it with minimal 
thought.  My mission is to take you to the next level where you know what 
to look for in a script, thus you can make sensible adjustments, even 
improvements to suit your circumstances. 

I realize that using time as the script s vehicle for the third week, may seem 
tedious to some.  However, the other side of the coin is that familiarity 
makes the scripts easier to concentrate on the new methods.  Additionally, 
there is a sense of building on success, which is not to be underestimated 
when writing scripts. 

I have flirted with scripting languages that were very strict in declaring 
variables, in comparison VBScript is pure anarchy. For a start, you don't 
have to declare variables at all.  If you do declare variables, then you can 
use any reasonable combinations of letters and numbers; for example, a 
string variable could be strServer, sServer, or you need not start with 's' at 
all.  

Children don't like anarchy, and so it is with scripters.  My advice is to 
impose rules when you name variables, but make them your rules not mine. 
The concept behind the Hungarian system for naming variables is that the 
first letters give you a clue whether the variable will be used for a string, a 
number, a time or other category of object. For example, strTimeSrv (string) 
intSleepLong (integer number). You may notice that my scripts invariably 
begin with the seemingly inconsequential, Option Explicit. This forces me to 
declare my variables, the benefit is that Option Explicit warns me if (when) I 
make a typo later in the script. 

This Week's Scripting Mission - Sub Routine 

Like a footnote in a book, I have added the restart code as sub routine at the 
end of the script.  Normally subroutines are used where you need to recycle 
the same code for multiple tasks. Thinking ahead to more complex scripts, 
there are time and space advantages to repeating a routine.  This is my 
reasoning for familiarizing you with the Sub Routine syntax and the 'Call 
Routine()' command. 

Example: To Restart the Windows Time Service 

Two Purposes of the Script 

The main learning point in this example is how to script a restart of the 
Windows Time Service.  Incidentally, if you just change the value for 
strService, then you can adapt the script to restart other services.  I advise 
you to begin, not with the code, but in the Administrative Tools, Services 
snap-in. What you are looking for is the precise name of the service that you 
wish to restart, for example is the time service called, W32tm, Windows 
Time or W32Time?  To my surprise, the answer was W32Time. 

The secondary learning point is to synchronize with a time server.  From a 
computing perspective, the script sets the time server, then it restarts the 
service, only then does it synchronize with the newly configured internet 
time source.   

Instructions for Restarting the Time Service 

This script is designed for Windows servers, but there is no reason why it 
should not work on an XP machine.  If you use uk.pool.ntp.org or time-
a.nist.gov as the time server, make sure that your machine has an internet 
connection. 

1. Copy and paste the example script below into notepad or use a 
VBScript editor. 

2. One advantage of a good script editor such as OnScript is that you 
can see the line numbers, which helps when you have to 
troubleshoot error messages. 

3. Save the file with a .vbs extension, for example: RestartTime.vbs  
4. Double click RestartTime.vbs, observe the cmd session and check 

the message box.  

'=============================================== 
' RestartTime.vbs 
' DATE: January 2006 Version 2.2 
' COMMENT: Script to Restart the Time service 
'============================================== 
Option Explicit 
Dim objShell 
Dim intShortSleep, intLongSleep, strService 
Dim strTimeSrv 

 

Set objShell = CreateObject("WScript.Shell") 

 

strService = "w32Time" 
intShortSleep = 500 
intLongSleep = 8000     '1000 = 1 second 

 

' Time Server set (Remove ' Rem if you want to change) 
'strTimeSrv = "time-a.nist.gov" 
strTimeSrv = "uk.pool.ntp.org" 

 

' Cmd prompt opened with .Run method 
objShell.Run "cmd"  
Wscript.Sleep intShortSleep 
objShell.SendKeys "{Enter}" 

 

' set the time server. Equivalent of net time /setsntp 
Wscript.Sleep intShortSleep 
objShell.SendKeys "w32tm /config /syncfromflags:manual 
/manualpeerlist:"_ 
& strTimeSrv  

 

Call Restart() 

 

' Point where the script synchronizes 
objShell.SendKeys "w32Tm /resync /rediscover" 
objShell.SendKeys "{Enter}" 
Wscript.Sleep intLongSleep 

 

' Cmd prompt exited 
objShell.SendKeys "Exit" 
Wscript.Sleep intShortSleep 
objShell.SendKeys "{Enter}" 

 

WScript.Quit  

 

Sub Restart() 
' Restart Service 
objShell.SendKeys "{Enter}" 
Wscript.Sleep intShortSleep 
objShell.SendKeys "net stop " & strService  
objShell.SendKeys "{Enter}" 
Wscript.Sleep intLongSleep  
objShell.SendKeys "net start " & strService  
objShell.SendKeys "{Enter}" 
Wscript.Sleep intLongSleep 
End Sub 

VBScript Tutorial - Learning Points 

Note 1:  Remember how scripts mimic commands that you can issue 
manually; in this case 'net stop serviceName' and 'net start serviceName'  

Note 2:  Option Explicit forces me to declare all my variables.  This is why I 
have Dim statements at the top of my scripts, to dimension variables such 
as strService. 

Note 3:  Pay close attention to how the script constructs the routine: Sub 
Restart().  This is not difficult, but you need to obey all the rules.  Start with 
the specific word: Sub.  Follow sub with the name of the routine and then 
end with brackets().  When you need to insert the routine into the main 
script, the command is Call Restart().  These are two common mistakes :a) 
Call Sub Restart() or b) Call Restart. 

Note 4:  Check the logic of the script, namely to:  
i) Set the Time Server 
ii) Restart the Windows Time Server 
iii) Only now do we, resync \ rediscover.  

Challenges 

 

Experiment with longer values for intLongSleep = 8000.  At 8 seconds, I 
may have left the value on the short side. 

 

Try removing the net stop command. 

 

Give the script a real job by manually stopping the Windows Time service 
before running the script. 

 

Seek other Services that you could restart, then strip down the script and 
adapt it to a service such as Alerter or Messenger.  

Summary of Restarting the Windows Time Service 

This script fits well with my master plan of building a script in stages.  The 
secret for easy learning is get the basic section working then bolt on other 
routines.  Another of my hobby-horses is that scripters have to be multi 
talented.  For this script you need knowledge of the Windows Services, the 
Net Stop command, topped off with a dose of logical thinking about the 
sequence of the operating system commands.  



 
Intro VBScript 1-10.doc  

Page 15 of 47    Author Guy Thomas  

Contents for Ezine 4 - 'If' Statements 

 
This Week's Missions

  
This Week's Secret

  
Example: To Synchronize with the Windows Time Service

  
Summary of If Statements

 

This Week's Missions - 'If' Statement also Datepart and Time 

This week's main mission is to master the tiny but useful, 'If' conditional 
statement.  Every scripting language has an 'If' construction, we are going to 
study the VBScript syntax for 'If'...Then  End If.  I will also touch on 
variations such as ElseIf.  

The context for this week's script is checking the system clock, (next week I 
promise we will use a new vehicle for our scripts - Active Directory).  What 
we want to do is display different messages depending on whether the 
computer's clock is fast, slow, or synchronized with the external time 
source.  Here is an example: 

If timeDiff < 0 then 
WScript.Echo "Clock was fast by " & -timeDiff & " secs" 
End If 

0) I like this 3 line version of the 'If' logic.  However, it is possible to employ 
short 'If' statements all on one line. 
    If timeDiff < 0 then xyx. 

1) Observe the word 'then'.  See how 'then' follows the test, 'If timeDiff < 0' 
and notice that 'then' is the last word on the first line. 

2) To tell VBScript that 'If' has finished evaluating the test statement, sign 
off with, 'End If' (note the space between End and If. 

3) As you can see, the 'If' construction is flexible.  In scripts where there is a 
main test and only one subsidiary, then I recommend that you employ 'If' 
(as usual) and just one plain 'Else' statement.  With VBScript, when you 
have three or four tests, then precede the second and subsequent test with 
ElseIf. 

ElseIf is different from Else If.  Where possible, choose ElseIf (all one word) 
because then you only need one End If no matter how many ElseIfs you 
have.  Avoid Else (space) If unless you have a specific reason for using this 
construction, the reason being each Else (space) If needs its own End If.  As 
ever, examples clarify thinking. 

Example A (Good) 

If timeDiff < 0 then  
     WScript.Echo "Clock was fast by " & -timeDiff & " secs" 
     ElseIf timeDiff > 0 then  
     WScript.Echo "Clock was slow by " & timeDiff & " secs" 
     ElseIf timeDiff = 0 then  
     WScript.Echo " Clock synchronized " & timeDiff & " difference" 
End if  

Below, I added the extra End If, caused by choosing Else If instead of ElseIf.  

Example B (Not Recommended) 

If timeDiff < 0 then  
     WScript.Echo "Clock was fast by " & -timeDiff & " secs" 
     Else If timeDiff > 0 then  
     WScript.Echo "Clock was slow by " & timeDiff & " secs" 
     Else If timeDiff = 0 then  
     WScript.Echo " Clock synchronized " & timeDiff & " difference" 
End if  
End if  
End if 

4) Should you have more than five ElseIf statements then consider a 
different construction altogether - Select Case. 

Time and Date Functions -  DATEPART and Time 

Almost incidentally, this week's script features time functions.  There is a 
whole family of related date and time functions, for example, NOW, TIME 
and DATEPART.  In this week's ezine, Datepart and TIME play a role  in 
calculating whether the computer clock was fast, slow or synchronized with 
the time server.  As with previous scripts, this script cries out for you to 
deliberately alter the computer's clock; otherwise the script just says 'Clock 
synchronized' . 

When Datepart reads the TIME function, "s" means seconds, minutes is 'n' 
(and not "m"), while hour is ("h").  My mission was to translate the clock 
time into seconds, then we can compare one pure number with another. 

This Week's Secret 

SENDKEYS is a useful method for leaning what the code is doing.  I also use 
.Sendkeys for troubleshooting what went wrong.  However, in the case of 
production scripts favour the .RUN method.  You can imagine the potential 
for disaster if you or a user switches windows after one .Sendkeys {} 
command and before the next.  At best the script would fail, at worst it could 
unleash a disastrous sequence of commands. 

Example: To Synchronize with the Windows Time Service 

Purposes of the Script 

The script synchronizes the local machine with an internet time server, then 
displays a message indicating if the internal clock was slow, fast or on time. 

Instructions for Synchronizing with the Windows Time Service 

This script is designed for Windows servers, but there is no reason why it 
should not work on an XP machine.  If you use uk.pool.ntp.org or time-
a.nist.gov as the time server, make sure that your machine has an internet 
connection. 

1. Copy and paste the example script below into notepad or use a 
VBScript editor. 

2. One advantage of a good script editor such as OnScript is that you 
can see the line numbers, which helps when you have to 
troubleshoot error messages. 

3. Save the file with a .vbs extension, for example: SynchTime.vbs  
4. Double click SynchTime.vbs, and check the clock synchronization in 

the message box. 

   

'=============================================== 
' VBScript Source File -- Created with XLnow OnScript 
' SynchTime.vbs 
' AUTHOR: Guy Thomas  
' COMPANY: Computer Performance 
' DATE: January 2006 Version 3.2 
' COMMENT: Script to synchronize with the Time service 
'===============================================
====== 
Option Explicit 
Dim objShell 
Dim intShortSleep, intLongSleep, strService 
Dim strTimeSrv, timeBefore, timeAfter, timeDiff  
Set objShell = CreateObject("WScript.Shell")  
strService = "w32Time" 
intShortSleep = 3000 
intLongSleep = 6000 '1000 = 1 second  

 

' Time Server set (Remove ' Rem if you want to change) 
strTimeSrv = "time-a.nist.gov" 
'strTimeSrv = "uk.pool.ntp.org"  

 

' Use .Run method to configure the time server 
objShell.Run "w32tm /config /syncfromflags:manual /manualpeerlist:"_ 
& strTimeSrv  
Call Restart()  
' Collect time before the script synchronizes 
timeBefore = DatePart("s" , Now) + DatePart("n" , Now) *60 
timeBefore = timeBefore + DatePart("h", Now) *3600  

 

' Key command to resynchronize with time server 
objShell.Run "w32Tm /resync /rediscover" 

Wscript.Sleep intShortSleep  
timeAfter = DatePart("s" , Now) + DatePart("n" , Now) *60 
timeAfter = timeAfter + DatePart("h", Now) *3600  

 
' Cosmetic section to display the clock adjustment 
timeDiff = (timeAfter - timeBefore) - (intShortSleep/1000) 
If timeDiff < 0 then  
    WScript.Echo "Clock was fast by " & -timeDiff & " secs" 
       ElseIf timeDiff > 0 then  
       WScript.Echo "Clock was slow by " & timeDiff & " secs" 
       ElseIf timeDiff = 0 then  
       WScript.Echo " Clock synchronized " & timeDiff & " difference" 
End if  
WScript.Quit  

 

Sub Restart() 
' Restart Service 
objShell.Run "net stop " & strService  
objShell.Run "net start " & strService  
Wscript.Sleep intLongSleep 
End Sub  

   

VBScript Tutorial - Learning Points 

Note 1:  Get into the rhythm of the If, then, End If.  This simple 
construction is versatile and greases the wheels of so many VBScripts. 

Note 2:  ElseIf is better than Else If.  My advice is avoid a space between 
Else and If, unless you deliberately want a different outcome.  If you have 
more than five ElseIfs then investigate the alternative Select Case. 

Note 3:  Observe how this script employs the .Run method rather than 
.Sendkeys {}. 

Challenges 

 
Try setting the computer's script fast or slow deliberately . 

 
Give the script a real job by stopping the Windows Time service before 

running the script. 

 
Experiment with the intSleep variables.  Try changing the values, or even 

removing them. 

 
See the difference a space makes in Else If (Instead of ElseIf) and see how 

Else If requires its own matching End If. 

 

I tried simplifying, timeAfter = DatePart("s" , Time) + DatePart("n" , Time) 
*60 
to timeAfter = Time.  My results were disappointing, the value was 
always the same, even though I changed the time on the computer 
clock. 

Summary of 'If' Statement 

The 'If' conditional statement is flexible and adds value to so many 
VBScripts.  Take the time to master the syntax, especially of ElseIf. 

SendKeys is great in testing and troubleshooting, but production scripts are 
more reliable if you employ the .Run method.  

Contents for Ezine 5 - Binding to Active Directory and Creating OUs 

 
This Week's Secret

  
This Week's Mission

  
Binding to Active Directory

  
Example Script: To Create a Top Level OU

  
Summary Bind to Active Directory and Create an OU

 

This Week's Secret 

This is week five of my series of ten newsletters.  Together they combine to 
give you an 'Introduction to VBScript'.  As I write my ezine, so I keep in 
mind a picture of my readers.   My twist this week is that I believe that most 
people read the ezine for period of 10 to 20 weeks, by which time they have 
either mastered VBScript or give up.  So this is why I concentrate on the 
basics of VBScript and why I produced this handy ebook. 

This Week's Mission 

Overall Mission  

Our overall mission is to control Active Directory objects such as Users, 
Computers and OUs (Organizational Units) with VBScript.  This week's 
skirmish is to understand how VBScript extracts LDAP names from Active 
Directory, a process known as binding.  It's appropriate that we use the 
container object OU as the vehicle for this week's script, because the OU can 
hold future test objects. 

My reason for selecting the OU for this week's script is twofold. 

1) Use Short Names for OUs 

When people name their OUs, they give little thought to scripting.  
Consequently they end up with OU names which make life difficult for script 
writers.  Either the names are too long, have spaces or contain non-alpha 
numeric characters.  Let me be clear, as far as Windows is concerned you 
can have 255 characters and all manner of characters and spaces.  It's only 
Guy the script writer who objects to more that 12 words (no spaces).  My 
point that learning scripting syntax is hard enough without introducing errors 
caused by spelling mistakes in the object's name. 

2) Favour Short Trees. 

I would want a good reason to have more than two levels of OUs.  Aside 
from introducing spelling errors, deep nesting causes sequencing problem 
when scripting OUs.  The rule is OU=Child,OU=Parent.  This may sound 
counter intuitive, until you remember that the full path to an object would 
be: 

cn=name, OU=Child, OU=Parent, DC=domain, DC=org.  Once you realise 
how LDAP constructs names, then the order for scripting Child / Parent OUs 
becomes obvious. 

So if you ignore my advice and have a three tier structure such as:  

Headquarters #1 /Child Section 4 /Grand Child Out.Reach 

Then the LDAP name is OU=Headquarters #1,OU=Child Section 4, 
OU=Grand Child Out Reach.  Woops I made a mistake, there is a missing full 
stop, it should be OU=Headquarters #1,OU=Child Section 4, OU=Grand 
Child Out.Reach.  Pedantic, yes; pathetic, possibly; difficult to detect the 
typo - certainly.  My point, if you are going to script, choose short punchy 
names with no spaces or punctuation. 

Binding to Active Directory 

It is vital to be able to connect to your domain name.  There are two ways of 
binding to Active Directory, the smart way and the staid, old-fashioned way.  
I love the smart way, because one script works on any domain, without 
alteration.  I dislike the old-fashioned way because it means you have to 
hard-code the domain name in the script, as I result I predict that 7 out of 
10 my readers will ignore the instruction to substitute their own domain for 
the name that I use in the script. 

Guy's Favoured Way to Bind to ANY active Directory 

  

' Section to bind to ANY Active Directory. 
Set objRoot = GetObject("LDAP://rootDSE") 
objDomain = objRoot.Get("defaultNamingContext") 
Set objDomain = GetObject("LDAP://" & objDomain) 

    

Note 1: Observe how we build the objDomain in stages.  First we GetObject, 
then we use the .Get method to retrieve the defaultNamingContext.  (Only 
later will we use the .Create method) 

Example Script: To Create a Top Level OU 

Instructions for Binding with Active Directory and Creating an OU 

This script is designed for Windows Active Directory.  You really need to run 
this script on a domain controller rather than an XP workstation.  

1. Copy and paste the example script below into notepad or use a 
VBScript editor. 

2. One advantage of a good script editor such as OnScript is that you 
can see the line numbers, which helps when you have to 
troubleshoot error messages. 

3. Save the file with a .vbs extension, for example: TopOU.vbs  
4. Double click TopOU.vbs, and check Active Directory Users and 

Computers for your new OU.   

' TopOU.vbs 
' VBscript to create an OU (Organizational Unit) 
' Note two steps to set domain 
' Author Guy Thomas http://computerperformance.co.uk/ 
' Version 1.5 - February 2006 
' ----------------------------------------------------------'  
Option Explicit 
Dim objRoot, objDomain, objOU 
Dim strContainer 

 

strContainer ="OU=Accounts" 

 

' Section to bind to ANY Active Directory. 
Set objRoot = GetObject("LDAP://rootDSE") 
objDomain = objRoot.Get("defaultNamingContext") 
Set objDomain = GetObject("LDAP://" & objDomain) 

 

' Section to create the OU defined by strContainer  
' Also note the use of: .SetInfo 

 

'On Error Resume next 
Set objOU = objDomain.Create("organizationalUnit", strContainer) 
objOU.SetInfo 

 

WScript.Echo "New Top Level OU created = " & strContainer 
WSCript.Quit 

   

Learning Points - Alternative Methods to Bind to Active Directory 

Binding Variation 1) Maximum Efficiency 

  

Set objRoot = GetObject("LDAP://rootDSE") 
Set ObjDomain = GetObject( "LDAP://" & 
objRoot.Get("defaultNamingContext")) 

    

I call Variation 1 the Maximum Efficiency method because it has only two 
lines and fewer instructions to achieve the binding.  This method is even 
better than my method, however I stick with my trusted 3-line method 
because I find it easier to understand and troubleshoot.  Hidden agenda, 
there are always at least 3 ways of scripting anything. 

Binding Variation 2) Dinosaur Method 

  

Set objDomain = GetObject("LDAP://dc=cp,dc=mosel") 

I call Variation 2 the Dinosaur method because it's set in its ways.  You have 
to hard-code the domain name, in this case dc=cp,dc=mosel.  Incidentally, 
dc means domain context and not domain controller.  If I use this variation 
in my example, I have to persuade readers to substitute their own domain 
for cp.mosel  (I can imagine 50% failing to read the instruction). 

To tell the truth, hard-coding is not a big problem in the real world, because 
once a script works for you domain, it will continue to work without further 
modification.  Variation 2 does have the advantage of only requiring one line. 

Learning Points - Creating the Organizational Unit (OU) 

Note 1:  See how the .Create method builds the OU with the name specified 
by strContainer. 
Set objOU = objDomain.Create("organizationalUnit", strContainer) 

Note 2:  Understand how the instruction: objOU.SetInfo, mimics you 
clicking the OK button.   Hidden agenda, remember how scripting merely 
reflects the way you click menus and buttons manually. 

Challenge 

Try and create a child OU.  Make a prediction, should it be: 
strContainer ="OU=child,OU=Accounts"  or strContainer 
="OU=Accounts,OU=child"  

Summary Binding to Active Directory and Creating OUs 

Scripting Active Directory object requires many skills.  This week we have 
concentrated on two concepts, binding to active directory, understanding the 
OU parent child relationship.  



 
Intro VBScript 1-10.doc  

Page 24 of 47    Author Guy Thomas  

Contents for Ezine 6 - Collect VBScript Methods 

 
This Week's Secret

  
This Week's Mission

  
Binding to Active Directory

  
Example: To Modify the Properties of Computer Objects

  
Summary -  Methods

 

This Week's Secret 

This week's secret is never miss a chance to collect a VBScript method.  In 
the scripting model, objects such as user or computer are important, 
however, it's the methods that convert plans into actions.  Next week we will 
focus on the LDAP properties, for example, givenName, sAMAccountName 
and displayName, but this week it's methods, such as .Put and .SetInfo that 
are caught in my spotlight. 

I think of methods as the doers, the men of action.  As I write my web 
pages, I search for the verb with just the correct shade of meaning, when I 
script I seek methods, which perform just the right action. 

This Week's Mission 

This week's mission is to modify the computer object's properties.  I choose 
the computer because it's often a neglected object, also I want to re-enforce 
the point that all Active Directory objects have more properties than you can 
imagine. 

My hidden agenda is to alert you to the presence of two properties with 
similar names, description and displayName.  As another aside, you can 
script information about the department to which your computers' belong. 

Example Script: To Modify the Properties of Computer Objects 

Pre- requisites 

You need an OU called Accounts, else change my value for strContainer. 

It makes sense to create or move two or three computer objects in the OU 
specified by the above strContainer. 

This script is designed for Windows Active Directory.  You really need to run 
this script on a domain controller rather than an XP workstation.  

Instructions for modifying the properties of computer objects. 

1. Copy and paste the example script below into notepad or use a 
VBScript editor. 

2. One advantage of a good script editor such as OnScript is that you 
can see the line numbers, which helps when you have to 
troubleshoot error messages. 

3. Save the file with a .vbs extension, for example: Description.vbs  
4. Double click Description.vbs, the check Active Directory Users and 

Computers for new values for your computer's properties. 
5. N.B. It may be necessary to right click the OU and select 'Refresh' 

from the drop down menu.  F5 is only works the first time then 
goes silent.   

  

' Description.vbs 
' VBscript with methods to modify computer properties 
' Author Guy Thomas http://computerperformance.co.uk/ 
' Version 1.2 - February 2006 
' ---------------------------------------------------------'  
Option Explicit 
Dim objRoot, objDomain, objOU, objComputer 
Dim strContainer, intCounter 

 

strContainer ="OU=Accounts," 

 

' Section to bind to ANY Active Directory. 
Set objRoot = GetObject("LDAP://rootDSE") 
objDomain = objRoot.Get("defaultNamingContext") 
strContainer = strContainer & objDomain 
set objOU =GetObject("LDAP://" & strContainer ) 
intCounter = 0 

 

' Start of Loop.  
' Note filter: If then...  
For each objComputer in objOU 
     If objComputer.class="computer" Then 
     objComputer.Put "Description", "Laptop" 
     objComputer.Put "DisplayName", "Toshiba" 
     objComputer.SetInfo 
     intCounter = intCounter +1 
     End if 
next 
WScript.Echo intCounter & " Descriptions changed to Laptop" 
WScript.Quit 

Learning Points  

Note 1:  The first method, .Put is like typing (or pasting) the values into the 
computer's property sheet; while the second method, .SetInfo is like 
pressing the OK button when you have finished modifying that object.  
Incidentally last week we employed the .Get and .Create methods.  

Get into the rhythm of modifying object with methods.  For example 
objComputer.Put, then the name of the property "Description", finally the 
value "Desktop". 

Note 2:  Following on from the previous ezine, this script binds to Active 
Directory then obtains the name of your domain, there is no need to hard-
code dc=xyz.  Nevertheless, do pay attention to strContainer, for instance 
this script requires a comma, after the name of the OU. 

Note 3:  Observe the simple but effective: For Each .... Next loop. 

Note 4:  This script uses the If ... then ... End If, to filter just computer 
objects. 

Challenges 

1) My Active Directory challenge is to add more properties, for example: 
objComputer.Put "Department", "Human Resources". 

2) My scripting challenge is to introduce more variables, for example: 
Dim strDescription 
strDescription = "Acme Computers" 
objComputer.Put "Description", strDescription  
(in place of: objComputer.Put "Description", "Desktop") 

Summary - Methods 

Never miss the chance to learn a new VBScript method.  Whenever you see 
a new script, by all means 'parse' it for the names of objects or properties, 
but in particular seek the methods.  For example .Put and .SetInfo.  

Contents for Ezine 7 - Create Users 

 
This Week's Secret

  
This Week's Mission

  
Example 1 Script: To Create a Basic User Object

  
Example 2 Script: To Create a 'Live' User

  
Summary -  Create User

 

This Week's Secret 

When you use VBScript to create major objects such as Users, Computer or 
Groups, your script must set a value for the LDAP property 
sAMAccountName.  Imagine, if you were creating a user manually, then 
sAMAccountName equates to 'User logon name'.  Also keep in mind Active 
Directory enforces that each sAMAccountName is unique in the domain, 
hence you get an error when you run the script for a second time (unless 
you change the value of strUser). 

Tip:  You can also employ sAMAccountName when you want to control the 
username attribute.  For example, suppose you want to map: profilePath \\ 
server \ users\ username; it is often easier to use the sAMAccountName, for 
example: profilePath \\ server \ users\ sAMAccountName. 

This Week's Mission 

Last week we modified the computer object.  This week we are going to 
create a user object.  My rationale for this learning sequence is that it's 
easier to modify than to create.  When we create an object, we must specify 
a minimum of three properties, whereas modify focuses only on one 
property. 

These are three mandatory properties needed to create a user: 
a) The type of object, for example user, and not computer, group or contact. 
b) CN= Set the common name. (N.B. not to be confused with CN = 
container.) 
c) sAMAccountName, the logon name.  Often set to the same value as CN. 

In the case of Users, it's optional, but helpful, if we give them a password 
and then enable their account by setting userAccountControl = 512.  As a 
finishing touch, we will set the account so that the user must change their 
password at first logon (PwdLastSet = 0). 

There are zillions more properties that we could configure, for example sn 
(Last Name), however that would get us side tracked from the main 
objective, which is just to create a user object.  I say just create, but it still 
gives me a thrill when VBScript delivers a user to the correct container. 

I have to also confess that normally, you create not one but lots of users.  
To achieve this, you would persuade the script to loop through the cells of a 
spreadsheet extracting the relevant values.  However, opening files is a 
whole different lesson and I want to concentrate on one topic at a time.  
Also, remember that last week we learned how to modify objects, so if the 
worst came to the worst, we could always run another script to add values to 
our users' property sheets. 

Example 1 Script: To Create a Basic User Object 

Pre- requisites 

You need an OU called Accounts, else change my value for strContainer. 

This script is designed for Windows Active Directory.  You really need to run 
this script on a domain controller rather than an XP workstation.  

Instructions for modifying the properties of computer objects. 

1. Copy and paste the example script below into notepad or use a 
VBScript editor. 

2. One advantage of a good script editor such as OnScript is that you 
can see the line numbers, which helps when you have to 
troubleshoot error messages. 

3. Save the file with a .vbs extension, for example: NewUser.vbs  
4. Double click NewUser.vbs, the check Active Directory Users and 

Computers for new values for your computer's properties. 
5. N.B. It may be necessary to right click the OU and select 'Refresh' 

from the drop down menu.  F5 is only works the first time then 
goes silent.   

  

' NewUser.vbs 
' Sample VBScript to create a User in Users (or strOU). 
' Author Guy Thomas http://Computerperformance.co.uk/ 
' Version 1.2 - February 2006 
' ------------------------------------------------------'  
Option Explicit 
Dim strUser, strOU 
Dim objRoot, objContainer, objUser 

 

'strOU = "cn=users," 
strOU = "OU=Accounts," 
strUser = "DomGuy97"  

 

' Bind to Active Directory, note: 'Set' and 'Get' 
Set objRoot = GetObject("LDAP://rootDSE") 
Set objContainer = GetObject("LDAP://" & strOU & _ 
objRoot.Get("defaultNamingContext"))  

 

' Build the actual "User". Note cn= 
' Observe the .Create method  
Set objUser = objContainer.Create("User", "cn=" & strUser) 
objUser.Put "sAMAccountName", strUser 
objUser.SetInfo 

 

WScript.Echo "Check " & strOU & " for " & strUser 
WScript.Quit  

 

' End of Sample Create Users VBScript. 

Learning Points  

Note 1:  Observe how the first part of the VBScript gets a handle on the OU 
and thus set the location where the user will be born.  Specifically, trace the 
role of: Set objRoot and GetObject. 

Note 2:  Examine how VBScript creates the type of object, "User" (not 
computer).  See how VBScript then names the new user with cn=strUser.  
Finally, the script adds the mandatory, sAMAccountName. 

Note 3:  Remember that .SetInfo is like pressing the OK button in the Active 
Directory Users and Computers dialog box. 

Note 4:  The WScript.Echo is not strictly necessary, however, my scripts are 
littered with echo commands which add indications that, a) Something has 
happened!  b) To remind us where to look for the new object. 

Example 2 Script: To Create a 'Live' User 

Did you notice that big red X next to the first user we created?   A red X 
indicates the account is disabled and could not logon. This script adds a 
password and then enables the user account.  As a final touch, objUser.Put 
"PwdLastSet", 0 means that the user must change the password at first 
logon. 

  

' CreateUsersAdv.vbs 
' Sample VBScript to create, and enable a User in strOU. 
' Author Guy Thomas http://Computerperformance.co.uk/ 
' Version 2.4 - February 2006 
' ------------------------------------------------------'  
Option Explicit 
Dim strUser, strOU, strPassword 
Dim objRoot, objContainer, objUser, objShell, objMMC 

 

'strOU = "cn=users," 
strOU = "OU=Accounts," 
strUser = "DomAdvGuy30"  
strPassword = "P@ssw0rd" 

 

' Bind to Active Directory. 
Set objRoot = GetObject("LDAP://rootDSE") 
Set objContainer = GetObject("LDAP://" & strOU & _ 
objRoot.Get("defaultNamingContext"))  

 

' Build the actual User. 
Set objUser = objContainer.Create("User", "cn=" & strUser) 
objUser.Put "sAMAccountName", strUser 
objUser.SetInfo 

 

' Section to set the password. Note second SetInfo 
objUser.SetPassword strPassword 
objUser.Put "userAccountControl", 512 
objUser.Put "PwdLastSet", 0 
objUser.SetInfo 

' Optional Subroutine to open ADUC 
Call MMC 

 
WScript.Quit  

 
sub MMC() 
set objMMC = CreateObject("WScript.Shell") 
objMMC.Run "%systemroot%\system32\dsa.msc" 
End Sub 

 

' End of Advanced Create Users VBScript. 

    

Learning Points  

Note 0:  Once a script gets to 20 lines, you really do benefit from a script 
editor such as OnScript. 

Note 1:  While 'SetPassword' is self evident, 'userAccountControl' is a topic 
in its own right.  Suffice to say that a vale of 512 means enable the account, 
while 514 means disable.  (However there are other values.) 

Note 2:  A value of zero for PwdLastSet, tells the user to reset their 
password at next logon. 

Note 3:  You may have noticed my recent penchant for sub routines, for 
example sub MMC().  Strategically, this fits with the modular design of my 
scripts; in this instance, 'call MMC' launches the Active Directory Users and 
Computers snap-in.  My hidden agenda is that .Run dsa.msc is a change 
from my usual WScript.echo. 

Challenges 

Research other LDAP properties, for example, sn and givenName, then add 
extra lines of script when you create a User object. 

Summary - Create Users 

A basic User only requires us to script three properties, the common name, 
(CN), the type of object, "User" and the sAMAccountName.  A more useful 
User object also has a password and is enabled ready for logon.  

Contents for Ezine 8 - Objects, Especially FSO 

 
This Week's Secret

  
This Week's Mission

  
Example 1 Script: Which Writes to a Text File

  
Example 2 Variation with a Sub Routine

  
Summary -  Objects

 

This Week's Secret 

All modern scripting languages are object based, so in this respect, VBScript 
is not exceptional.  However, I am still in awe of the way that VBScript can 
call upon Windows Script Host for an endless supply of objects.  To me, each 
type of object has its own personality. This is because each object has 
different properties and is manipulated by different methods.  Here are the 5 
main types of VBScript object: 

1. Active Directory, User, Group or Computer (LDAP//:domainame) 
2. File System Object (FSO) read or write to text files 
3. Network (As in network drive or printer) 
4. Shell (As in Windows Shell - Run a program such as Explorer) 
5. WMI (Disk, memory, event viewer)   

My secret is that I do not study these objects types out of a desire for 
theoretical knowledge, but because I forget which object has which 
capabilities.  For example, I was trying to be flashy and launch the Windows 
Explorer; I tried:  
set objExp = CreateObject("WScript.Network") 
objExp.Run ("Explorer" & " " & strPath ) 

Can you see what's wrong?  CreateObject("WScript.Network"), should be, 
CreateObject("WScript.SHELL").  I foolishly choose an object with the wrong 
properties.  The truth is I solved the problem by doing what you would have 
done, I searched for an old script where I knew the .Run method had 
worked.  Once I saw that the object was .Shell it was easy to correct my 
mistake. 

This Week's Mission 

This week's mission is to examine the variety of VBScript object types, and 
then hone in on one type: File System Object (FSO).  To recap, objects have 
properties and we control what the object does with methods.  To take an 
example, from this week's script, we can append data to a File System 
Object with the .writeline method. 

What I want to do is get you started with the simplest FSO script, namely 
just to write data into a file.  This technique is particularly handy when you 
need to record the results in a permanent file.  One case in point is a WMI 
script, where you do not want to echo the data to the screen, instead you 
want to store the computer data on disk. 

Other jobs for FSO include reading from spreadsheets, for example, if you 
are building zillions of users from data held in Excel. 

This Week's Mission 

This week's mission is to examine the variety of VBScript object types, and 
then hone in on one type: File System Object.  To recap, objects have 
properties and we control what the object does with methods.  To take an 
example, from this week's script, we can append data to a FSO (File System 
Object) with the .writeline method. 

What I want to do is get you started with the simplest FSO script, namely 
just to write data into a file.  This technique is particularly handy when you 
need to record the results in a permanent file.  One case in point is  a WMI 
script, where you do not want to echo the data to the screen, instead you 
want to store the computer data on disk. 

Other jobs for FSO include reading from spreadsheets, for example, if you 
are building zillions of users from data held in Excel. 

Example 1 Script: To Write to a Text File 

Pre- requisites 

You need a folder to hold the file.  I choose E:\ Files.  Therefore, to make 
sure your script works, either create a folder called 'Files' on your 'e: drive', 
or better still, change the value of strPath to reflect your preferred folder. 

This script should work on any Windows, machine, this week you do not 
need Active Directory. 

Instructions for modifying the properties of computer objects. 

1. Copy and paste the example script below into notepad or use a 
VBScript editor. 

2. One advantage of a good script editor such as OnScript is that you 
can see the color codes for the methods and values. 

3. Save the file with a .vbs extension, for example: FSO.vbs  
4. Double click FSO.vbs, the check the contents of the file specified by 

strFile.   

' FSO.vbs 
' VBScript to write to a file 
' Author Guy Thomas http://Computerperformance.co.uk/ 
' Version 1.2 - February 2006 
' ------------------------------------------------------'  
Option Explicit 
Dim objFile, objExp 
Dim strPath, strFile, strFilePath, strGuyFile 

  

strPath = "e:\files\" 
strFile = "Guy.txt" 
strFilePath = strPath & strFile 

 

' FSO creates the object called - objFile 
Set objFile = CreateObject("Scripting.FileSystemObject") 
Set strGuyFile = objFile.CreateTextFile(strFilePath, True) 

 

' Method called .WriteLine to add data. 
strGuyFile.WriteLine("This was made using an FSO Object.") 
strGuyFile.Close 

 

' Bonus section to actually open the file 
set objExp = CreateObject("WScript.Shell") 
objExp.Run (strFilePath ) 

 

WScript.Quit 

Learning Points  

Note 1:  This script creates a File System Object and not a network or Active 
Directory object.  The script then puts objFile to work and as a result, a text 
file is born.  We control the name of this file with strGuyFile. 

Note 2:  The situation is this, we have created an empty file.  Now we select 
the .WriteLine method to enter the data into the file, just as though we had 
typed the letters on the keyboard. 

Note 3:  It is interesting that in order to open the Windows Explorer, we call 
for a second type of object; we select a Shell objects because it has the .Run 
method.  Neither FileSystemObjects nor network objects support the .Run 
method. 

Example 2 Write to a file.  Variation with Sub Routine. 

Just a small variation on the Write to File method; this example employs a 
sub routine to launch the Windows Explorer. 

' FSO_Wex.vbs 
' VBScript to write to a file 
' Author Guy Thomas http://Computerperformance.co.uk/ 
' Version 1.3 - February 2006 
' ------------------------------------------------------'  
Option Explicit 
Dim objFile, objExp 
Dim strPath, strFile, strFilePath, strGuyFile 

 

strPath = "e:\files\" 
strFile = "GuySub.txt" 
strFilePath = strPath & strFile 
Set objFile = CreateObject("Scripting.FileSystemObject") 
Set strGuyFile = objFile.CreateTextFile(strFilePath, True) 
strGuyFile.WriteLine("This used Sub Explorer().") 
strGuyFile.Close 

 

Call Explorer 

 

WScript.Quit 

 

' Sub routine to launch Windows Explorer 
sub Explorer() 
set objExp = CreateObject("WScript.Shell") 
objExp.Run ("Explorer" & " " & strPath ) 

 

End Sub 

Learning Points  

Note 0:  Invest in OnScript and see how color coding makes your scripts 
easier to follow and to troubleshoot. 

Note 1:  The only real point of the second script is to demonstrate my latest 
craze of using sub routines.  In truth, sub routines are only beneficial if you 
reuse the code later in the script.  My personal indulgence is to use sub 
routines in constructing complex scripts from a series of modules or sub 
routines. 

Challenges 

1. Experiment with other values for strPath.   
2. How about adding a variable to control the text that you write in the 

file?  For example, declare strText, then control its value with strText = 
"This was made using an FSO Object."  Finally, amend 
strGuyFile.WriteLine(strText) 

Summary - Objects 

VBScript can create an endless stream of objects. Each object has its own 
properties and thus is suitable for certain scripting tasks, but not others. File 
System Object is useful addition to script where you need to store the output 
in a text file. 

Contents for Ezine 9 - VBScript Arguments 

 
This Week's Secret

  
This Week's Mission

  
Example 1: To Map a Network Drive to letter U:

  
Example 2: Employ extra arguments for MapNetworkDrive

  
Summary -  MapNetworkDrive

 

This Week's Secret 

In everyday life I try to avoid arguments.  It's much the same with VBScript, 
I only look for extra arguments as a last resort. 

Regarding logon scripts, my position is that I love logon scripts, they are my 
best selling ebook.  Moreover, I am happy to help people troubleshoot their 
logon scripts. 

The big 'however', is that I wonder whether Logon Scripts have a long-term 
future (6-10 years).  It is my belief that one day a Group Policy will do 
everything that a logon script can do today.  Therefore my advice in 2006 is 
this, check to see if there is Group Policy to achieve what you want e.g. 
Folder Direction for the home drive, if not then by all means use a logon 
script. 

My final word on this debate is that if you deploy logon scripts, make sure 
that you apply the scripts via Group Policy and not via the user's property 
sheet. 

This Week's Mission 

Last week we created scripts using the FileSystemObject, this week we are 
going to use the Network object.  Our mission is to investigate the 
MapNetworkDrive method and in due course, deploy all five of its 
arguments.  Incidentally, VBScript's methods are not case sensitive, the 
capitalization is merely a way of making its meaning easier to decipher. 

Example 1  MapNetworkDrive 

This example is designed as both an introduction for beginners and as a 
refresher for the more experienced scripters.  This example only employs the 
two commonest arguments, drive letter and UNC path. 

Example 2  MapNetworkDrive -  Arguments 

Example 2 features the thrust of this week's ezine, to illustrate that VBScript 
methods often have extra arguments.  In this example we employ all five 
MapNetworkDrive arguments. 

Example 1: To Map a Network Drive to letter U: 

Pre- requisites 

You need a free local drive letter.  Rather eccentrically, I have chosen letter 
U as the drive letter because my H: is already in use.  Feel free to amend 
strDriveLetter. 

You will need a shared folder.  If no other machine is available, share out a 
folder on your own machine.  Make sure that you amend strRemotePath to 
reflect your UNC path and not mine. 

This script should work on any Windows, machine.  However, Example 2 will 
need active directory for the username. 

Instructions for mapping network drive 

1. Copy and paste the example script below into notepad or use a 
VBScript editor. 

2. One advantage of a good script editor such as OnScript is that it 
displays each method's arguments. 

3. Save the file with a .vbs extension, for example: 
mapnetworkdrive.vbs  

4. Double click mapnetworkdrive.vbs, the check the drive letters with 
Windows Explorer.   

  

' MapNetworkDrive.vbs 
' VBScript to map a network drive.  
' Author Guy Thomas http://computerperformance.co.uk/ 
' Version 1.2 - March 2006 
' -------------------------------------------------------------'  
Option Explicit 
Dim objNetwork, objExp 
Dim strDriveLetter, strRemotePath 

 

' Values of variables set 
strDriveLetter = "u:"  
strRemotePath = "\\grand\ezine"  

 

' This section creates a network object. (objNetwork) 
' Then applies the MapNetworkDrive method. Result U: drive 

 

Set objNetwork = CreateObject("WScript.Network")  
objNetwork.MapNetworkDrive strDriveLetter, strRemotePath 

 

' Just to check the script has executed. 
WScript.Echo "Check explorer for " & strDriveLetter 
WScript.Quit 

 

' End of Example script . 

Learning Points  

Note 1:  Example 1 illustrates the classic VBScript technique: object, 
method, value.  This week we feature a network object (not a 
FileSystemObject). 

Note 2:  In a production script we would remove the WScript.echo 
command.  In testing, I like such commands because they demonstrate that 
the script has completed. 

Note 3:  This script, while great for learning the MapNetworkDrive basics, 
does have deficiencies.  For instance, it does not allow you to specify a 
different user's credentials.  More irritatingly, it gives an error message if 
you run the script for a second time.  For now, add : 
On Error Resume Next.  Next week will deal properly with this error problem. 

Example 2: Employ extra arguments for MapNetworkDrive 

We really need a real life scenario to explain why you would add these three 
arguments, strUser, strPassword, bUpdateProfile.  The first reason is 
security.  Often administrators want to help users to map drives, but they do 
not want to give them rights to the root of the network share.  The answer is 
to employ arguments that embed an administrator's name and password in 
the script.  To digress, the name does not have to be 'The Administrator', if 
you are going to do this for real, you could create a generic user and only 
give them permissions on the share you are mapping.  

To digress even further, I can hear the security conscious amongst you 
saying, 'shouldn't we encrypt the password'; well I have investigated, but I 
have not found a way of encrypting the password within VBScript.  If it is 
possible, then a kind reader will let me know and I will post the method here 
for all to see. 

The other argument is bUpdateProfile.  When I was a newbie, the letter 'b' in 
bUpdateProfile surprised and confused me, then one day I realized that it 
was 'b' for Boolean, meaning it only took two values, true or false.  This 
relatively trivial argument controls whether - or not, to update the users 
profile.  Incidentally, the default is false. 

Pre- requisites 

It is essential to change the strUser and strPassword otherwise this script 
will not work on your network. 

' MapNetWorkDriveA5.vbs 
' VBScript to map a network drive with 5 arguments.  
' Author Guy Thomas http://computerperformance.co.uk/ 
' Version 1.4 - March 2006 
' --------------------------------------------------------------'  
Option Explicit 
Dim objNetwork, objExp 
Dim strDriveLetter, strRemotePath, strUser, strPassword, bUpdateProfile 

 

' Values of variables set 
strDriveLetter = "V:"  
strRemotePath = "\\grand\ezine"  
strUser = "thomasg"   'N.B. Must be valid username 
strPassword = "P#ssw0rd"    'Must be correct password 
bUpdateProfile = "true" 

 

' This section creates a network object. (objNetwork) 
' Then apply MapNetworkDrive method. Result V: drive 
' Note, this script features 5 arguments on lines 24/25. 

 

Set objNetwork = CreateObject("WScript.Network")  

 

objNetwork.MapNetworkDrive strDriveLetter, strRemotePath, _ 
bUpdateProfile, strUser, strPassword  

 

' Calls subroutine to check if drive is already connected 
call Explorer 

 

WScript.Quit 

 

' Sub routine to launch Windows Explorer 
sub Explorer() 
set objExp = CreateObject("WScript.Shell") 
objExp.Run ("Explorer" & " " & strDriveLetter ) 

 

End Sub 

 

' End of MapNetworkDrive Example script . 

    

http://computerperformance.co.uk/

Learning Points  

Note 0:  When you use a script editor such as OnScript, it helps you by 
displaying a Method's arguments. 

Note 1:  This is the core script of this week's ezine.  Here is where we 
employ all five of MapNetworkDrive's arguments.  In particular we enable 
MapNetworkDrive to execute as thought you entered a different user's name 
and password in the manual dialog box.  (For example, Windows Explorer, 
Tools menu, Map Network Drive). 

Note 2:  Observe how we use the underscore (_).   
The reason is because all five arguments won't fit on one line. 

Note 3:  I replaced the WScript.echo with a sub routine which opens 
Windows Explorer, the result is that you can see the mapped drive easily. 

Note 4: The script still lacks error correcting code, consequently, when you 
run the script a second time, you get error 80070055. Next week we will 
employ logic to check if the drive is already in use; as a temporary measure, 
we could add 
'On Error Resume Next' and thus prevent the annoying error. 

Summary - MapNetworkDrive Arguments 

This week we are dealing with network objects.  Many VBScript methods 
have well known argument, but often there are lesser know optional 
arguments.  The focal point of this ezine is Example 2; here is where we 
employ all five of the MapNetworkDrive methods to create the Mapped 
Network Drive. 

Contents for Ezine 10 - Error Correcting Code 

 
This Week's Secret

  
This Week's Mission

  
Example 1: Basic Error Correcting Code

  
Example 2: Researching err.number

  

Example 3: Adding Err.Number

  

Summary Error Correcting Code

 

This Week's Secret 

One of my best skills is anticipation, perhaps the result of playing chess in 
my youth.  One of my worst skills is paying attention to detail.  Therefore, 
when it comes to troubleshooting VBScript, I have the skill to find the 
problems, but lack the desire to add error correcting code.  I admit that I 
also shelter under the excuse that error correcting code makes my example 
scripts too tortuous and therefore, more difficult to understand.  This week I 
want to compensate for this omission and provide strategies for adding error 
correcting code to your VBScripts. 

This Week's Mission 

This week's mission is to add error correcting code to your (or to my) 
scripts.  Let me ask this question: 'What happens to a script without extra 
code to trap errors?'  The answer is nothing, provided the conditions are 
ideal.  However, if there is an unexpected exception, then the script is likely 
to halt with a WSH error box.  For example, the script tries to map a drive 
letter that is already in use. 

To add comprehensive error correcting code takes at least twice the effort 
you invested in creating the original script.  Therefore, in the case of 'run of 
the mill' scripts, I add error correcting techniques that are merely 'good 
enough'. 

Back to anticipation.  What I do is scan all the possible areas which could 
cause my lovely script to fail.  Wrong syntax, faulty logic, uncontrolled 
exceptions e.g. null values, server not available.  Then if it's an important 
script, I go the extra mile and add error correcting code. 

In the real world, you get additional problems caused by psychotic users, 
who do things ordinary mortals could not imagine.  Also, in some 
circumstances, there may be security concerns with hackers using your 
scripts for unintended purposes.  

To digress, where I come from in Wales, it's common to be referred to by 
your occupation, for example, Evans the Fish, or Jones the Coke (sold Coca 
Cola).  In school my friend was always known as Davies Dead, hid dad was 
an undertaker. 

Guy the Ruthless. (On Error Resume Next) 

The first thing 'Guy the Ruthless' does is add Option Explicit at the beginning 
of each VBScript.  This forces me to declare all variables, and thus avoid 
careless errors.  For example, if I begin by calling a variable objNetwork, but 
then refer to it as strNetwork, Option Explicit warns me that I have not 
declared strNetwork.  Thus preventing me making a logical error later in the 
script. 

If I still find my script gives a problem, then I crudely and ruthlessly, add On 
Error Resume Next.  When you employ On Error Resume Next, it is best is to 
add this command to the line immediately before the error. 

By the way, even 'Guy the Ruthless' tests the script, this throws up any silly 
syntax errors.  What transforms 'Guy the Ruthless' from an average 
scriptwriter to a good script writer, is a top-notch script editor.  Investigate a 
program such as OnScript which identifies the line numbers and color codes 
objects, methods and string variables.  Get your evaluation copy of 
OnScript

 

Guy the Conscientious 

The first thing 'Guy the Conscientious' does is remind himself of Murphy's 
Law; what can go wrong will go wrong. This jolts me into anticipation mode 
where I ask myself, 'what if'.  What if the drive letter is already in use?  
What if there is no value in the 
spreadsheet's cell. 

Guy the Conscientious extends 
testing; he deliberately makes the 
script fail, for instance, by trying 
to map a drive to a letter he 
knows is already in use.  He then 
records the error message and 
writes code to 'handle' the halted 
script.   

The technique involves a special VBScript object called 'err'.  Err has a 
property called number, therefore the first stage when looking for clues is to 
add this line: 
WScript.echo "Error code is " & err.number.    

We hope that the result will be zero, meaning no error.  However, we must 
prepare for the worst, if the 'device is already in use, then you get an 
err.number such as -2147024811.  The key connection is realizing that this 
pure decimal number has a hex equivalent of 80070055.  The eureka 
moment come when you realize that this is the very same number which is 
displayed in the WSH error message box.  You can induce the error by 
deliberately running the script for a second time (Remove On error resume 
next).   

If you want to test at home launch the scientific calculator, click the decimal 
button and paste: 
-2147024811, next, click the hex button on the scientific calculator, 
expected result (FFFFFFFF) 80070055. 

Guy the Conscientious makes a special study of these hex numbers, which 
begin with 800x0000.  If you analyse the hex error message, you get a 
valuable clue depending on whether x (fourth digit) = 4,5, 7 or A.  For 
example,  5 = ADSI, 7 = logical, A= Syntax.  Therefore the 8007xxxx series 
are logical errors such as drive in use or server not available; they are not 
the fault of VBScript errors. 

Guy the Logical 

If you take error correcting code to the highest level, you need to combine 
knowledge of the hex and decimal codes with If ...then.. else... End If logic.  
In other words, so you have a problem - what are you going to about it?  For 
example, if the drive is already connected, then you could run a sub-routine 
to disconnect it.  I have to say that if you go this extra mile and error 
correcting code, you always get that feeling of satisfaction. 

Sometimes you get lucky and VBScript has an 'If Exist' command, for 
example, with File Scripting Object: 
If objFSO.FolderExists(strDirectory).  Unfortunately, MapNetworkDrive does 
not have an If Exist construction, therefore we are going to learn error 
correcting code from the bottom up. This week I am going to show you how 
to analyze the err.numbers.  Next week we concentrate on putting the logic 
into practice. 

Example 1: Basic Error Correcting Code 

This week, I want I want to build up gradually.  Therefore we start with a 
script which just map's the drive.  Later, in Example 2, I will modify the 
script to echo the problem.  Then in Example 3, I am going to develop error 
correcting logic and also indulge my passion for Select Case, rather than use 
a succession of 'if else' statements. 

Pre- requisites 

You need a free local drive letter.  Rather eccentrically, I have chosen letter 
U as the drive letter because my H: is already in use.  Feel free to amend 
strDriveLetter. 

You will need a shared folder.  If no other machine is available, share out a 
folder on your own machine.  Make sure that you amend strRemotePath to 
reflect your UNC path and not mine. 

This script should work on any Windows, machine.   

Instructions for mapping a network drive (Preparation Script) 

1. Copy and paste the example script below into notepad or use a 
VBScript editor. 

2. One advantage of a good script editor such as OnScript is that it 
displays a methods arguments.Get your evaluation copy of 
OnScript

 

3. Save the file with a .vbs extension, for example: BasicOnErr.vbs  
4. Double click BasicOnErr.vbs, the check the drive letters with 

Windows Explorer, note any WSH 800x0000 error messages. 

' BasicOnErr.vbs 
' VBScript to demonstrate error correcting code 
' Authors Guy Thomas 
' Version 1.4 - March 2006 
' ----------------------------------------------------------'  
'  
Option Explicit 
Dim objNetwork, strDrive, objShell 
Dim strRemotePath, strDriveLetter 
'  
strDriveLetter = "u:"  
strRemotePath = "\\grand\ezine"  

 

On Error Resume Next 

 

Set objNetwork = CreateObject("WScript.Network")  
objNetwork.MapNetworkDrive strDriveLetter, strRemotePath  

 

WScript.Quit 

 

' End of script. 

Learning Points  

Note 1:  Get into the habit of using Option Explicit at the start of your 
scripts.  What Option Explicit does is force you to declare all your variables 
before you use them.  The benefit is it saves typos in variable names. 

Note 2:  The only real error correcting code is the crude, but effective, On 
Error Resume Next.  What this does is permit the script to continue instead 
of halting with a WSH error message box.   Naturally it does not so much 
cure the problem, as sweep it under the carpet. 

Note 3:  Not demonstrated here, but if you wished to reverse On Error 
Resume Next, add 'On Error GoTo 0'.  N.B. Type: GoTo not Go To, and that's 
a zero not an oh. 

Example 2: Researching err.number 

Here is a script which is designed for research.  This is the middle stage in 
developing error correcting code.  The technique is to do things that 
deliberately make the script go wrong.  For example, change the server 
name and thus induce an error by making the code try to map a non-
existent UNC path.   

  

' ErrorCorrecting.vbs 
' VBScript to demonstrate error correcting code 
' Authors Guy Thomas 
' Version 2.4 - March 2006 
' ------------------------------------------------'  
'  
Option Explicit 
Dim objNetwork, strDrive, objShell 
Dim strRemotePath, strDriveLetter 
'  
strDriveLetter = "u:"  
strRemotePath = "\\grand\ezine"  
err.Clear 

 

On Error Resume Next 

 

Set objNetwork = CreateObject("WScript.Network")  
objNetwork.MapNetworkDrive strDriveLetter, strRemotePath  

 

if err.number <> 0 then WScript.Echo "Error number " & err.number 

 

WScript.Quit(0) 

 

' End of script. 

Learning Points  

Note 1:  The whole point of this script is to introduce my favourite 
troubleshooting technique, adding WScript.Echo at key points in a fault 
script.  Here we echo the err.number.  Once we have one, or preferably 
more err.numbers then we progress to Example 3. 

Example 3: Adding Err.Number 

Hidden away on line 27, Err.Number is the star command of this script.  
What it does is expose the reason the script failed.  In fact, err.number is 
only the first part of the error correcting processes.  In a real life script you 
would then make the instructions fork and take corrective action.  For 
example, if the drive letter was already in use (-2147024811) then you could 
run a sub routine to disconnect the mapped network drive.  Such a sub-
routine is not included in this script as we are building up gradually.  Next 
week will explore taking logical action based on error, but this week I want 
to concentrate on researching with err.number. 

Instructions 

Here is a golden opportunity to make the script go wrong by deliberately 
trying to map to a non existent UNC path, or to drive letters that are already 
in use. 

  
' ErrorCorrecting.vbs 
' VBScript to demonstrate error correcting code 
' Authors Guy Thomas 
' Version 2.3 - March 2006 
' ------------------------------------------------'  
'  
Option Explicit 
Dim objNetwork, strDrive, objShell 
Dim strRemotePath, strDriveLetter 
'  
strDriveLetter = "u:"  
strRemotePath = "\\grand\ezine"  
err.Clear 

 

On Error Resume Next 

 

Set objNetwork = CreateObject("WScript.Network")  
objNetwork.MapNetworkDrive strDriveLetter, strRemotePath  

 

if err.number <> 0 then call ECCode 
' Clear error else error number carries over and script fails 
Err.clear 

 

WScript.Quit(0) 

 

sub ECCode() 
Select Case err.number 
Case -2147024811 WScript.Echo "Drive in use" 
Case -2147023693 WScript.Echo "No Server" 
Case -2147942453 WScript.Echo "Network PATH not found" 
Case -2147942467 WScript.Echo "Network NAME not found" 
Case 0 WScript.Echo " ok" 
Case else Wscript.Echo "Investigate " & err.number 
End Select 
WScript.Quit(1) 
End Sub  

 

' End of script. 

Learning Points  

Note 0:  When you use a script editor such as OnScript, it helps you by 
displaying line numbers and color coded commands. Get your evaluation 
copy of OnScript

 
Note 1:  Investigate sub ECCode.  Understand how Select Case replaces 
multiple If then.... ElseIf statements.  Check the rhythm of each line, Case 
number (or condition), action. 

Note 2: Trace where the main script calls the sub routine.  Pay attention to 
where you need brackets; yes when you define sub ECCode(), no where you 
simply call ECCode. 

Challenges 

My challenge is open ended.  Decide what level of error correcting code your 
scripts need.  Experiment, then test, various logic routines to handle 
unexpected situations.   

Summary - Error Correcting Code 

This week I wanted to concentrate on error correcting strategies. Tactics 
such as 'Option Explicit' and 'On Error Resume next', are easy to understand 
and implement.  However, adding true error correcting code based on 
err.number is time consuming.  The final piece in the error correcting jigsaw 
is logic. 

Next week I will develop the error logic so that it actually takes corrective 
action and thus handle errors usefully and gracefully.   

